# $Id: PKGBUILD 71778 2012-06-02 10:30:24Z bluewind $
# Maintainer: Sergej Pupykin <pupykin.s+arch@gmail.com>
# Maintainer: Tom K <tomk@runbox.com>

pkgname=perl-fuse
pkgver=0.14
pkgrel=2
pkgdesc="write filesystems in Perl using FUSE"
arch=('i686' 'x86_64')
url="http://search.cpan.org/dist/Fuse"
depends=('perl' 'fuse')
license=('GPL')
source=(http://www.cpan.org/authors/id/D/DP/DPAVLIN/Fuse-$pkgver.tar.gz)
md5sums=('df72f17bf03e1a31c1a834816a3a59f1')

build() {
  cd  $srcdir/Fuse-$pkgver
  PERL_MM_USE_DEFAULT=1 perl Makefile.PL INSTALLDIRS=vendor
  make
}

package() {
  cd  $srcdir/Fuse-$pkgver
  make install DESTDIR=$pkgdir
  find $pkgdir -name '.packlist' -delete
  find $pkgdir -name '*.pod' -delete
}
